1 Introduction

We consider the one-electron Hydrogenic-atom Hamiltonian, which is of the form

$$\hat{H}\left|\psi\right\rangle = E\left|\psi\right\rangle$$

where $\hat{H} = \hat{K} + \hat{V}$, with

$$\hat{K}\left|\psi\right\rangle = \left[-\frac{1}{2r}\frac{\partial^2}{\partial r^2}\big(r\cdot\big) + \frac{1}{2r^2}\hat{L}^2\right]\left|\psi\right\rangle \quad \text{and} \quad \hat{V}\left|\psi\right\rangle = -\frac{Z}{r}\left|\psi\right\rangle$$

and where $\hat{L} = \hat{L}_x + \hat{L}_y + \hat{L}_z$ is the angular momentum operator, which has eigenstates $\left|Y_\ell^m\right\rangle$ which satisfy

$$\hat{L}^2\left|Y_\ell^m\right\rangle = \ell(\ell+1)\left|Y_\ell^m\right\rangle \quad \text{and} \quad \hat{L}_z\left|Y_\ell^m\right\rangle = m\left|Y_\ell^m\right\rangle.$$

We solve this system by the method of basis expansion, where we utilise a basis of the form, $\mathcal{B} = \{\left|\phi_i\right\rangle\}_{i=1}^N$ which we suppose to be complete in the limit as $N \to \infty$. We select the basis functions, represented in coordinate-space, to be of the form

$$\phi_i(r,\Omega) = \frac{1}{r}\varphi_{k_i,\ell_i}(r)Y_{\ell_i}^{m_i}(\Omega) \quad \text{for} \quad i = 1,\ldots,N$$

where the radial functions, $\mathcal{R} = \{\left|\varphi_{k_i,\ell_i}\right\rangle\}_{i=1}^N$ form a complete basis for the radial function space, in the limit as $N \to \infty$. For elements of this basis, the one-electron Hydrogenic-atom Hamiltonian assumes the form

$$\begin{aligned}
\hat{H}\left|\phi_i\right\rangle &= \left[-\frac{1}{2r}\frac{\partial^2}{\partial r^2}\big(r\cdot\big) + \frac{1}{2r^2}\hat{L}^2 - \frac{Z}{r}\right]\left|\phi_i\right\rangle \\
&= \left[-\frac{1}{2r}\frac{\partial^2}{\partial r^2}\big(r\cdot\big) + \frac{\ell_i(\ell_i+1)}{2r^2} - \frac{Z}{r}\right]\left|\phi_i\right\rangle \\
&= \left[-\frac{1}{2r}\frac{\partial^2}{\partial r^2}\big(r\cdot\big) + \frac{\ell_i(\ell_i+1)}{2r^2} - \frac{Z}{r}\right]\left|\tfrac{1}{r}\varphi_{k_i,\ell_i}, Y_{\ell_i}^{m_i}\right\rangle
\end{aligned}$$

thus reducing to operator which acts purely to radial terms, indexed by ℓ_i. Lastly, we note that the inner product is of the form

$$\left\langle\phi_i\right|\hat{A}\left|\phi_j\right\rangle = \int_0^\infty \mathrm{d}r\, r^2 \int_\Omega \mathrm{d}\Omega\, \overline{\phi_i(r,\Omega)}\hat{A}\big[\phi_j(r,\Omega)\big]$$

where $\hat{A}$ is an arbitrary linear operator, and whence, in the case where $\hat{A}$ can be reduced to an operator which acts only on radial terms, indexed by ℓ, we have that

$$\begin{aligned}
\left\langle\phi_i\right|\hat{A}\left|\phi_j\right\rangle &= \int_0^\infty \mathrm{d}r\, r^2\overline{\tfrac{1}{r}\varphi_{k_i,\ell_i}(r)}\hat{A}_{\ell_j}\left[\tfrac{1}{r}\varphi_{k_j,\ell_j}(r)\right]\int_\Omega \mathrm{d}\Omega\,\overline{Y_{\ell_i}^{m_i}(\Omega)}Y_{\ell_j}^{m_j}(\Omega) \\
&= \int_0^\infty \mathrm{d}r\, r^2\overline{\tfrac{1}{r}\varphi_{k_i,\ell_i}(r)}\hat{A}_{\ell_j}\left[\tfrac{1}{r}\varphi_{k_j,\ell_j}(r)\right]\delta_{\ell_i,\ell_j}\delta_{m_i,m_j} \\
&= \left\langle\tfrac{1}{r}\varphi_{k_i,\ell_i}\right|\hat{A}_{\ell_j}\left|\tfrac{1}{r}\varphi_{k_j,\ell_j}\right\rangle\delta_{\ell_i,\ell_j}\delta_{m_i,m_j}
\end{aligned}$$

where we have defined the radial inner product to be of the form

$$\left\langle\tfrac{1}{r}\varphi_{k_i,\ell_i}\right|\hat{A}_{\ell_j}\left|\tfrac{1}{r}\varphi_{k_j,\ell_j}\right\rangle = \int_0^\infty \mathrm{d}r\, r^2\overline{\tfrac{1}{r}\varphi_{k_i,\ell_i}(r)}\hat{A}_{\ell_j}\left[\tfrac{1}{r}\varphi_{k_j,\ell_j}(r)\right].$$

2 Laguerre Basis

We utilise a Laguerre basis for the set of radial functions which, for $k = 1, 2, \ldots$ and where $\ell \in \{0, 1, \ldots\}$, are of the following form in coordinate-space

$$\varphi_{k,\ell}(r) = N_{k,\ell}(2\alpha_\ell r)^{\ell+1} \exp(-\alpha_\ell r) L_{k-1}^{2\ell+1}(2\alpha_\ell r)$$

where $\alpha_\ell \in (0, \infty)$ is an arbitrarily chosen constant, where $N_{k,\ell}$ are the normalisation constants, given by

$$N_{k,\ell} = \sqrt{\frac{\alpha_\ell (k-1)!}{(k+\ell)(k+2\ell)!}}$$

and where $L_{k-1}^{2\ell+1}$ are the generalised Laguerre polynomials.

2.1 Recurrence Relation

We construct the Laguerre basis by means of the following recurrence relation of the Laguerre polynomials

$$\begin{aligned} L_0^t(x) &= 1 \\ L_1^t(x) &= 1 + t - x \\ (n+1)L_{n+1}^t(x) &= (2n + 1 + t - x)L_n^t(x) - (n+t)L_{n-1}^t(x) \quad \text{for} \quad n = 1, 2, \ldots. \end{aligned}$$

Firstly, we write $\varphi_{k,\ell}(r) = N_{k,\ell}\widetilde{\varphi}_{k,\ell}(r)$, whence we note that

$$\begin{aligned} \widetilde{\varphi}_{1,\ell}(r) &= (2\alpha_\ell r)^{\ell+1} \exp(-\alpha_\ell r) \\ \widetilde{\varphi}_{2,\ell}(r) &= 2(\ell + 1 - \alpha_\ell r)(2\alpha_\ell r)^{\ell+1} \exp(-\alpha_\ell r) \\ (k-1)\widetilde{\varphi}_{k,\ell}(r) &= 2(k - 1 + \ell - \alpha_\ell r)\widetilde{\varphi}_{k-1,\ell}(r) - (k + 2\ell - 1)\widetilde{\varphi}_{k-2,\ell}(r) \quad \text{for} \quad k = 3, 4, \ldots, \end{aligned}$$

from which can trivially recover the functions $\varphi_{k,\ell}(r)$.

2.2 Normalisation Constant Recurrence Relation

To circumvent overflow errors when calculating the normalisation constant, $N_{k,\ell}$, we construct these constants using a recurrence relations. We note that

$$\begin{aligned} N_{k,\ell} &= \sqrt{\frac{\alpha_\ell (k-1)!}{(k+\ell)(k+2\ell)!}} \\ &= \sqrt{\frac{(k-1)(k-1+\ell)}{(k+\ell)(k+2\ell)} \frac{\alpha_\ell (k-2)!}{(k-1+\ell)(k+2\ell-1)!}} \\ &= \sqrt{\frac{(k-1)(k-1+\ell)}{(k+\ell)(k+2\ell)}} N_{k-1,\ell} \end{aligned}$$

for $k = 2, 3, \ldots$ and where $\ell \in \{0, 1, \ldots\}$, and that

$$N_{1,\ell} = \sqrt{\frac{\alpha_\ell}{(\ell+1)(2\ell+1)!}}$$

yielding a numerically-stable recurrence relation for the normalisation constants as required.

2.3 Laguerre Radial Basis Code

FORTRAN code for calculating the Laguerre basis functions for a given radial grid can be found in `src/laguerre.f90: subroutine radial_basis()`, and is shown in Listing 1.

```
! radial_basis
!
! phi_{k, l, m}(r, theta, phi) = (varphi_{k, l}(r) / r) * Y_{l, m}(theta, phi)
! where
! varphi_{k, l}(r) = sqrt(alpha * (k - 1)! / (k + l) * (k + 2*l)!)
!                    * (2*alpha*r)^{l+1}
!                    * exp(-alpha*r)
!                    * L_{k - 1}^{2*l + 1}(2*alpha*r)
! where L_{i}^{j} are the generalised Laguerre polynomials.
!
! For given l, alpha, and r_grid, yields the functions varphi_{k, l}(r) for
! k = 1, ..., n_basis, on the radial values specified in the grid.
!
! Also returns an error code <ierr> where:
! - 0 indicates successful execution,
! - 1 indicates invalid arguments.
pure subroutine radial_basis (l, alpha, n_r, r_grid, n_basis, basis, ierr)
  integer , intent(in) :: l, n_r, n_basis
  double precision , intent(in) :: alpha
  double precision , intent(in) :: r_grid(n_r)
  double precision , intent(out) :: basis(n_r, n_basis)
  integer , intent(out) :: ierr
  double precision :: norm(n_basis)
  double precision :: alpha_grid(n_r)
  integer :: kk

  ! check if arguments are valid
  ierr = 0

  if ((l < 0) .or. (n_basis < 1) .or. (n_r < 1)) then
    ierr = 1
    return
  end if

  ! recurrence relation for basis normalisation constants
  norm(1) = sqrt(alpha / dble((l + 1) * gamma(dble((2 * l) + 2))))

  if (n_basis >= 2) then
    do kk = 2, n_basis
      norm(kk) = norm(kk-1) * sqrt(dble((kk - 1) * (kk - 1 + l)) / &
          dble((kk + l) * (kk + (2 * l))))
    end do
  end if

  ! in-lined array since r_grid(:) on its own is never used
  alpha_grid(:) = alpha * r_grid(:)

  ! recurrence relation for basis functions
  basis(:, 1) = ((2.0d0 * alpha_grid(:)) ** (l + 1)) * &
      exp(-alpha_grid(:))

  if (n_basis >= 2) then
    basis(:, 2) = 2.0d0 * (dble(l + 1) - alpha_grid(:)) * basis(:, 1)
```

```
    end if

    if (n_basis >= 3) then
      do kk = 3, n_basis
        basis(:, kk) = &
            ((2.0d0 * (dble(kk - 1 + l) - alpha_grid(:)) * basis(:, kk-1)) &
            - dble(kk + (2 * l) - 1) * basis(:, kk-2)) / dble(kk - 1)
      end do
    end if

    ! scaling basis functions by normalisation constants
    do kk = 1, n_basis
      basis(:, kk) = basis(:, kk) * norm(kk)
    end do

  end subroutine radial_basis
```

Listing 1: Calculation of Laguerre radial basis functions for a given radial grid.

2.4 Laguerre Radial Basis Figures

A radial grid has been constructed, for given $\mathrm{d_r}$ and $\mathrm{r_{max}}$, of the form

$$\{r_i = \mathrm{d_r} \cdot (i-1)\}_{i=1}^{\mathrm{n_r}}$$

where $\mathrm{n_r}$ is the smallest integer such that

$$\mathrm{d_r} \cdot (\mathrm{n_r} - 1) \geq \mathrm{r_{max}}.$$

The Laguerre basis functions have been calculated on this radial grid, for various values of ℓ and α_ℓ. The plots of the first 4 basis functions for these values of ℓ and α_ℓ are shown in Figure 1.

It can be seen from Figure 1 that as ℓ increases, the Laguerre radial functions are somewhat shifted further from the origin; that is, they are suppressed at the origin, peak at a further distance away from the origin, and extend further away from the origin before exponentially decaying to 0. They also have wider and less pronounced peaks.

It can also be seen from Figure 1 that as α_ℓ decreases, the Laguerre radial basis functions extend much further away from the origin before exponentially decaying to 0, and have wider and less pronounced peaks.



(a) $\varphi_{k,0}(r)$ for $k = 1, 2, 3, 4$, with $\alpha_\ell = 1.0$.

(b) $\varphi_{k,1}(r)$ for $k = 1, 2, 3, 4$, with $\alpha_\ell = 1.0$.

(c) $\varphi_{k,0}(r)$ for $k = 1, 2, 3, 4$, with $\alpha_\ell = 0.5$.

(d) $\varphi_{k,1}(r)$ for $k = 1, 2, 3, 4$, with $\alpha_\ell = 0.5$.

Figure 1: The first four Laguerre radial basis functions are plotted for various cases of ℓ and α_ℓ. Note that every figure has the same y-axes bounds $[-0.75, 0.75]$, whereas the x-axes bounds are $[0, 20]$ for the $\alpha_\ell = 1.0$ cases, and $\{0, 30\}$ for the $\alpha_\ell = 0.5$ cases. Observe that $\varphi_{k,\ell}(r)$ has k extremal points, with each extrema being larger (in magnitude) than the preceeding extrema, before eventually exhibiting exponential decay to 0, after the last extremal point.

3 Kinetic Energy Matrix Elements

Here, we shall derive an analytic expression for the kinetic energy matrix elements, with regard to the Laguerre radial basis. Firstly, we note that the kinetic energy matrix elements are defined by

$$K_{i,j} = \langle \phi_i | \hat{K} | \phi_j \rangle = \langle \phi_i | \left[-\frac{1}{2r} \frac{\partial^2}{\partial r^2} (r\cdot) + \frac{1}{2r^2} \hat{L}^2 \right] | \phi_j \rangle$$

whence we note that

$$\begin{aligned} K_{i,j} &= \left\langle \tfrac{1}{r}\varphi_{k_i,\ell_i}, Y_{\ell_i}^{m_i} \right| \left[-\frac{1}{2r} \frac{\partial^2}{\partial r^2} (r\cdot) + \frac{1}{2r^2} \hat{L}^2 \right] \left| \tfrac{1}{r}\varphi_{k_j,\ell_j}, Y_{\ell_j}^{m_j} \right\rangle \\ &= \left\langle \tfrac{1}{r}\varphi_{k_i,\ell_i} \right| \left[-\frac{1}{2r} \frac{\mathrm{d}^2}{\mathrm{d}r^2} (r\cdot) + \frac{\ell_j(\ell_j+1)}{2r^2} \right] \left| \tfrac{1}{r}\varphi_{k_j,\ell_j} \right\rangle \left\langle Y_{\ell_i}^{m_i} \middle| Y_{\ell_j}^{m_j} \right\rangle \\ &= \left\langle \tfrac{1}{r}\varphi_{k_i,\ell_i} \right| \hat{K}_{\ell_j} \left| \tfrac{1}{r}\varphi_{k_j,\ell_j} \right\rangle \delta_{\ell_i,\ell_j} \delta_{m_i,m_j}. \end{aligned}$$

We note that since the matrix element is necessarily zero, where $\ell_i \neq \ell_j$, we restrict our attention to the case where $\ell_i = \ell_j = \ell$. It follows that the radial terms can be written in the form

$$\begin{aligned} \left\langle \tfrac{1}{r}\varphi_{k_i,\ell} \right| \hat{K}_{\ell} \left| \tfrac{1}{r}\varphi_{k_j,\ell} \right\rangle &= \left\langle \tfrac{1}{r}\varphi_{k_i,\ell} \right| \left[-\frac{1}{2r} \frac{\mathrm{d}^2}{\mathrm{d}r^2} (r\cdot) + \frac{\ell(\ell+1)}{2r^2} \right] \left| \tfrac{1}{r}\varphi_{k_j,\ell} \right\rangle \\ &= \int_0^\infty \mathrm{d}r\, r^2 \overline{\tfrac{1}{r}\varphi_{k_i,\ell}(r)} \left[-\frac{1}{2r} \frac{\mathrm{d}^2}{\mathrm{d}r^2} (r\cdot) + \frac{\ell(\ell+1)}{2r^2} \right] \left(\tfrac{1}{r}\varphi_{k_j,\ell}(r) \right) \\ &= \int_0^\infty \mathrm{d}r\, \varphi_{k_i,\ell}(r) \left[-\frac{1}{2} \frac{\mathrm{d}^2}{\mathrm{d}r^2} + \frac{\ell(\ell+1)}{2r^2} \right] \varphi_{k_j,\ell}(r) \end{aligned}$$

where we have dropped the conjugacy due to the Laguerre radial basis functions being entirely real-valued. Expanding this fully, we have that

$$\begin{aligned} \left\langle \tfrac{1}{r}\varphi_{k_i,\ell} \right| \hat{K}_{\ell} \left| \tfrac{1}{r}\varphi_{k_j,\ell} \right\rangle &= \int_0^\infty \mathrm{d}r \left(N_{k_i,\ell} (2\alpha r)^{\ell+1} \exp(-\alpha r) L_{k_i-1}^{2\ell+1}(2\alpha r) \right) \\ &\quad \times \left[-\frac{1}{2} \frac{\mathrm{d}^2}{\mathrm{d}r^2} + \frac{\ell(\ell+1)}{2r^2} \right] \left(N_{k_j,\ell} (2\alpha r)^{\ell+1} \exp(-\alpha r) L_{k_j-1}^{2\ell+1}(2\alpha r) \right) \end{aligned}$$

whence we introduce the variable transformation $x = 2\alpha r$, to yield an equivalent integral of the form

$$\begin{aligned} \left\langle \tfrac{1}{r}\varphi_{k_i,\ell} \right| \hat{K}_{\ell} \left| \tfrac{1}{r}\varphi_{k_j,\ell} \right\rangle &= (2\alpha) N_{k_i,\ell} N_{k_j,\ell} \int_0^\infty \mathrm{d}x\, x^{\ell+1} \exp(-\tfrac{x}{2}) L_{k_i-1}^{2\ell+1}(x) \\ &\quad \times \left[-\frac{1}{2} \frac{\mathrm{d}^2}{\mathrm{d}x^2} + \frac{\ell(\ell+1)}{2x^2} \right] \left(x^{\ell+1} \exp(-\tfrac{x}{2}) L_{k_j-1}^{2\ell+1}(x) \right). \end{aligned}$$

At this point, we note that

$$\begin{aligned} \frac{\mathrm{d}^2}{\mathrm{d}x^2} \left(x^{\ell+1} \exp(-\tfrac{x}{2}) L_{k_j-1}^{2\ell+1}(x) \right) &= x^{\ell+1} \exp(-\tfrac{x}{2}) \\ \times \Bigg(\left(\frac{\ell(\ell+1)}{x^2} - \frac{\ell+1}{x} + \frac{1}{4} \right) L_{k_j-1}^{2\ell+1}(x) &+ \left(\frac{2(\ell+1)}{x} - 1 \right) \frac{\mathrm{d}}{\mathrm{d}x} \left(L_{k_j-1}^{2\ell+1}(x) \right) + \frac{\mathrm{d}^2}{\mathrm{d}x^2} \left(L_{k_j-1}^{2\ell+1}(x) \right) \Bigg) \end{aligned}$$

whence

$$\left[-\frac{1}{2}\frac{\mathrm{d}^2}{\mathrm{d}x^2}+\frac{\ell(\ell+1)}{2x^2}\right]\big(x^{\ell+1}\exp(-\tfrac{x}{2})L^{2\ell+1}_{k_j-1}(x)\big) = -\frac{1}{2}x^{\ell+1}\exp(-\tfrac{x}{2}) \times\left(\left(-\frac{\ell+1}{x}+\frac{1}{4}\right)L^{2\ell+1}_{k_j-1}(x)+\left(\frac{2(\ell+1)}{x}-1\right)\frac{\mathrm{d}}{\mathrm{d}x}\big(L^{2\ell+1}_{k_j-1}(x)\big)+\frac{\mathrm{d}^2}{\mathrm{d}x^2}\big(L^{2\ell+1}_{k_j-1}(x)\big)\right).$$

We utilise the following recurrence relation of the generalised Laguerre polynomials,

$$\frac{t+1-x}{x}\frac{\mathrm{d}}{\mathrm{d}x}\big(L^t_n(x)\big)+\frac{\mathrm{d}^2}{\mathrm{d}x^2}\big(L^t_n(x)\big) = -\frac{n}{x}L^t_n(x)$$

to further simplify the above term to the form

$$\left[-\frac{1}{2}\frac{\mathrm{d}^2}{\mathrm{d}x^2}+\frac{\ell(\ell+1)}{2x^2}\right]\big(x^{\ell+1}\exp(-\tfrac{x}{2})L^{2\ell+1}_{k_j-1}(x)\big) = \frac{1}{2}\left(\frac{k_j+\ell}{x}-\frac{1}{4}\right)x^{\ell+1}\exp(-\tfrac{x}{2})L^{2\ell+1}_{k_j-1}(x)$$

whence the integral becomes

$$\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\right|\hat{K}_\ell\left|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle = \alpha N_{k_i,\ell}N_{k_j,\ell}\int_0^\infty \mathrm{d}x\left(k_j+\ell-\frac{x}{4}\right)x^{2\ell+1}\exp(-x)L^{2\ell+1}_{k_i-1}(x)L^{2\ell+1}_{k_j-1}(x)$$

We note that

$$\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\middle|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle = \frac{N_{k_i,\ell}N_{k_j,\ell}}{2\alpha}\int_0^\infty \mathrm{d}x\, x^{2\ell+2}\exp(-x)L^{2\ell+1}_{k_i-1}(x)L^{2\ell+1}_{k_j-1}(x)$$

whence the previous integral can be separated as

$$\begin{aligned}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\right|\hat{K}_\ell\left|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle &= \alpha N_{k_i,\ell}N_{k_j,\ell}(k_j+\ell)\int_0^\infty \mathrm{d}x\, x^{2\ell+1}\exp(-x)L^{2\ell+1}_{k_i-1}(x)L^{2\ell+1}_{k_j-1}(x)\\ &\quad-\frac{\alpha}{4}N_{k_i,\ell}N_{k_j,\ell}\int_0^\infty x^{2\ell+2}\exp(-x)L^{2\ell+1}_{k_i-1}(x)L^{2\ell+1}_{k_j-1}(x)\\ &= \alpha N_{k_i,\ell}N_{k_j,\ell}(k_j+\ell)\int_0^\infty \mathrm{d}x\, x^{2\ell+1}\exp(-x)L^{2\ell+1}_{k_i-1}(x)L^{2\ell+1}_{k_j-1}(x)\\ &\quad-\frac{\alpha^2}{2}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\middle|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle.\end{aligned}$$

At this point we note the following property of the generalised Laguerre polynomials,

$$\int_0^\infty \mathrm{d}x\, x^t\exp(-x)L^t_n(x)L^t_m(x) = \frac{(n+t)!}{n!}\delta_{m,n}$$

whence the radial term of the kinetic energy matrix elements is shown to be given analytically by the expression

$$\begin{aligned}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\right|\hat{K}_\ell\left|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle &= \alpha N^2_{k_i,\ell}(k_j+\ell)\frac{(k_j+2\ell)!}{(k_j-1)!}\delta_{k_i,k_j}-\frac{\alpha^2}{2}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\middle|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle\\ &= \alpha^2\delta_{k_i,k_j}-\frac{\alpha^2}{2}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\middle|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle.\end{aligned}$$

It follows that the kinetic energy matrix elements are thus of the form

$$K_{i,j} = \alpha^2\Big(\delta_{k_i,k_j}-\frac{1}{2}\left\langle \tfrac{1}{r}\varphi_{k_i,\ell}\middle|\tfrac{1}{r}\varphi_{k_j,\ell}\right\rangle\Big)\delta_{\ell_i,\ell_j}\delta_{m_i,m_j}.$$

3.1 Extension: Overlap Matrix Elements

4 Atomic Hydrogen States

4.1 Hydrogenic Atom Code

4.1.1 Overlap Matrix Elements

FORTRAN code for calculating the overlap matrix elements for a Laguerre radial basis of a given dimension can be found in src/laguerre.f90: subroutine overlap_matrix(), and is shown in Listing 2.

```
! overlap_matrix
!
! < phi_{k', l, m} | phi_{k, l, m} >
!
! Overlap matrix elements for given l, m.
! We can restrict our attention to considering fixed l and m, since the matrix
! elements are zero when l' /= l or where m' /= m.
! Furthermore, the exponential decay variable, alpha, has no influence on
! these matrix elements, nor does the magnetic quantum number, m.
!
! Also returns an error code <ierr> where:
! - 0 indicates successful execution,
! - 1 indicates invalid arguments.
pure subroutine overlap_matrix(l, m, n_basis, B, ierr)
  integer , intent(in) :: l, m, n_basis
  double precision , intent(out) :: B(n_basis, n_basis)
  integer , intent(out) :: ierr
  integer :: kk

  ! check if arguments are valid
  ierr = 0

  if ((l < 0) .or. (n_basis < 1)) then
    ierr = 1
    return
  end if

  ! initialise overlap matrix to zero
  B(:, :) = 0.0d0

  ! determine tri-diagonal overlap matrix elements
  do kk = 1, n_basis-1
    B(kk, kk) = 1.0d0

    B(kk, kk+1) = - 0.5d0 * sqrt(1 - &
        (dble(l * (l + 1)) / dble((kk + l) * (kk + l + 1))))

    B(kk+1, kk) = B(kk, kk+1)
  end do

  ! last term (not covered by loop)
  B(n_basis, n_basis) = 1.0d0

end subroutine overlap_matrix
```

Listing 2: Calculation of overlap matrix elements for a Laguerre radial basis of a given dimension.

4.1.2 Kinetic Energy Matrix Elements

FORTRAN code for calculating the kinetic energy matrix elements for a Laguerre radial basis of a given dimension can be found in `src/laguerre.f90: subroutine kinetic_matrix()`, and is shown in Listing 3.

```
  ! kinetic_matrix
  !
  ! < phi_{k', l, m} | K | phi_{k, l, m} >
  !
  ! Kinetic matrix elements for given l, m, alpha.
  ! We can restrict our attention to considering fixed l and m, since the matrix
  ! elements are zero when l' /= l or where m' /= m.
  ! Furthermore, the magnetic quantum number, m, has no influence on these
  ! matrix elements.
  !
  ! Also returns an error code <ierr> where:
  ! - 0 indicates successful execution,
  ! - 1 indicates invalid arguments.
  pure subroutine kinetic_matrix(l, m, alpha, n_basis, K, ierr)
    integer , intent(in) :: l, m, n_basis
    double precision , intent(in) :: alpha
    double precision , intent(out) :: K(n_basis, n_basis)
    integer , intent(out) :: ierr
    integer :: kk

    ! check if arguments are valid
    ierr = 0

    if ((l < 0) .or. (n_basis < 1)) then
      ierr = 1
      return
    end if

    ! initialise kinetic matrix to zero
    K(:, :) = 0.0d0

    ! determine tri-diagonal kinetic matrix elements
    do kk = 1, n_basis-1
      K(kk, kk) = 0.5d0 * (alpha ** 2)

      K(kk, kk+1) = (alpha ** 2) * 0.25d0 * sqrt(1 - &
          (dble(l * (l + 1)) / dble((kk + l) * (kk + l + 1))))

      K(kk+1, kk) = K(kk, kk+1)
    end do

    ! last term (not covered by loop)
    K(n_basis, n_basis) = 0.5d0 * (alpha ** 2)

  end subroutine kinetic_matrix
```

Listing 3: Calculation of kinetic energy matrix elements for a Laguerre radial basis of a given dimension.

4.1.3 Coulomb Potential Matrix Elements

FORTRAN code for calculating the Coulomb potential matrix elements for a Laguerre radial basis of a given dimension can be found in `src/laguerre.f90: subroutine coulomb_matrix()`, and is shown in Listing 4.

```
! coulomb_matrix
!
! < phi_{k', l, m} | 1/r | phi_{k, l, m} >
!
! Coulomb matrix elements for given l, m, alpha.
! We can restrict our attention to considering fixed l and m, since the matrix
! elements are zero when l' /= l or where m' /= m.
! Furthermore, the magnetic quantum number, m, has no influence on these
! matrix elements.
!
! Also returns an error code <ierr> where:
! - 0 indicates successful execution,
! - 1 indicates invalid arguments.
pure subroutine coulomb_matrix(l, m, alpha, n_basis, V, ierr)
  integer , intent(in) :: l, m, n_basis
  double precision , intent(in) :: alpha
  double precision , intent(out) :: V(n_basis, n_basis)
  integer , intent(out) :: ierr
  integer :: kk

  ! check if arguments are valid
  ierr = 0

  if ((l < 0) .or. (n_basis < 1)) then
    ierr = 1
    return
  end if

  ! initialise coulomb matrix to zero
  V(:, :) = 0.0d0

  ! determine diagonal coulomb matrix elements
  do kk = 1, n_basis
    V(kk, kk) = alpha / dble(kk + l)
  end do

end subroutine coulomb_matrix
```

Listing 4: Calculation of Coulomb potential matrix elements for a Laguerre radial basis of a given dimension.

4.1.4 Hamiltonian Matrix Elements

FORTRAN code for calculating the overlap, kinetic energy, potential energy, and Hamiltonian matrix elements for a Laguerre radial basis of a given dimension can be found in `src/laguerre.f90: subroutine hydrogenic_matrices()`, and is shown in ??.

```
! hydrogenic_matrices
!
! Yields overlap, kinetic, potential and Hamiltonian matrices for given l, m,
! alpha, atomic_charge; that is: B, K, V, H.
```

```
  ! We can restrict our attention to considering fixed l and m, since the matrix
  ! elements are zero when l' /= l or where m' /= m.
  ! Furthermore, the magnetic quantum number, m, has no influence on these
  ! matrix elements.
  !
  ! Also returns an error code <ierr> where:
  ! - 0 indicates successful execution,
  ! - 1 indicates invalid arguments.
  pure subroutine hydrogenic_matrices(l, m, alpha, atomic_charge, n_basis, B, &
      K, V, H, ierr)
    integer, intent(in) :: l, m, atomic_charge, n_basis
    double precision, intent(in) :: alpha
    double precision, intent(out) :: B(n_basis, n_basis), K(n_basis, n_basis), &
        V(n_basis, n_basis), H(n_basis, n_basis)
    integer, intent(out) :: ierr

    ! check if arguments are valid
    ierr = 0

    if ((l < 0) .or. (n_basis < 1)) then
      ierr = 1
      return
    end if

    ! calculate matrices
    call overlap_matrix(l, m, n_basis, B, ierr)

    call kinetic_matrix(l, m, alpha, n_basis, K, ierr)

    call coulomb_matrix(l, m, alpha, n_basis, V, ierr)
    V(:, :) = - dble(atomic_charge) * V(:, :)

    H(:, :) = K(:, :) + V(:, :)

  end subroutine hydrogenic_matrices
```

Listing 5: Calculation of overlap, kinetic energy, potential energy, and Hamiltonian matrix elements for a Laguerre radial basis of a given dimension.

4.2 Extension: He^+ Ion

4.3 Extension: Surface Plot in xz Plane

4.4 Extension: Numerically Calculating Potential Matrix Elements